ADDENDUM

TO

FLOPPY DISK CONTROLLER
USERS' MANUAL

FOR

MINIFLOPPY™ APPLICATIONS

TM - Shugart Associates

Five Militia Drive/Lexington, Massachusetts 02173 Telephone (617) 862-6410 Telex 92-3434

TABLE OF CONTENTS

INTRODUCTION

The discussion and design shown in this Addendum is dedicated to the Minifloppy™ Diskette Storage Drive. This drive is somewhat different, both electrically and mechanically, from the standard floppy disk drives found throughout the industry. The Minifloppy is smaller, less expensive, and is based upon the same floppy disk drive technology as the standard floppy. However, its data transfer rate is half as fast and its total data storage capacity is approximately one-third.

This design uses a single uPD372 Floppy Disk Controller to control a pair of Minifloppies. The interface to the floppies has been configured so as to allow overlap-seeks to be programmed. If only a single drive is used or if overlap-seeks are not required, then two or three logic IC's may be removed from this design. The controller's architecture remains the same as in the standard floppy design shown in the uPD372 Users' Manual.

Before proceeding into the design of this controller, the uPD372 Users' Manual should be read. All the basic concepts and characteristics of the uPD372 are explained in this document, and it will be assumed that the reader is familiar with them. For clarity in this document the uPD372 Users' Manual will be referred to as Users' Manual and this document referred to as the Addendum.

TM -- Shugart Associates

MINIFLOPPY INTERFACE SIGNALS

There are several signals which have been deleted in the Minifloppy as well as several new signals. These are summarized below:

Deleted Signals (Used only on Standard Floppy)	New Signals
Head Load	Index/Sector
Write Fault	Drive Select 1
Write Fault Reset	Drive Select 2
Low Current	Drive Select 3
Sector	Motor ON
Index	
Ready	

The Head Load signal has been deleted in the Minifloppy. The head is loaded concurrently with the Motor ON signal. Write Fault and Write Fault Reset which were tests of the Floppy's status prior to attempting to write a diskette, have both been eliminated. Many standard floppies (IBM compatible) have a Low Current signal which allows the write current in the recording head to be decreased on tracks 44-76 -- this signal has been eliminated. If a hard sector recording format is used, the standard floppies provided separate signal outputs for both Sector and Index. The Minifloppy requires that the user separate these signals (this is usually done with a one-shot circuit). The READY command in the standard floppy indicated to the controller that a diskette had been inserted, the door was closed, and that the diskette was spinning; this signal has also been eliminated.

Three separate device select lines are provided on the Minifloppy and the appropriate one is selected by the use of hardware straps in the drive. This allows a maximum of three drives to be selected without additional decoding hardware. The Minifloppy uses a dc motor for rotating the diskette, a separate signal called MOTOR ON is used for turning the motor on. In order to increase the longevity of the motor, software has been incorporated in the controller so that two seconds after the last program instruction, the motor is shut off.

The following figure shows a typical interface connection between the controller and the Minifloppy.



Figure 1 -- Controller/Minifloppy Interface

From the detail schematic (Figure 6) it can be seen that the functions of the following signals coming out of the uPD372 have been changed to execute the new commands.

Standard Floppy Command	372 Pin Number	Minifloppy Command
(WFR) Write Fault Reset	23	Motor ON (Device #1)
(LCT) Low Current	22	Motor ON (Device #2)

The functional performance of these signals is changed in software and requires no change to the uPD372 hardware.

The uPD372 has two pairs of device select lines, an A pair and a B pair. In this application these lines have been configured to control only two Minifloppies. The A pair (UAØ and UA1) are used as follows:

UAØ - Select Read/Write Electronics in Drive #1

UA1 - Select Read/Write Electronics in Drive #2

The B pair (UBØ and UB1) are used as follows:

UBØ - Select Motion Control Electronics in Drive #1

UB1 - Select Motion Control Electronics in Drive #2

Software constraints have been incorporated so that:

- Only one Drive may be Reading or Writing at a time.
- Only one Drive may be Stepping IN or OUT at a time.

However, it is possible to Read or Write on one Drive while stepping on the other. This is done by selecting UAØ • UB1 or UA1 • UBØ.

When two Minifloppies are used, connector J1 on the controller should be connected to Drive #1 and J2 to Drive #2, (Radial busing to the Drives). This should be done in order to keep the hardware decoding on IC's U5Ø, U51, U53 and U17 the same as the software listing.

RECORDING FORMAT

The software listing shown at the end of this Addendum and the discussion which follows is for a Soft Sectored format. However, if the user wishes to use Hard Sectoring, he may do so by simply changing the software (the uPD372 is not a limiting factor to hard sectoring). Soft Sectoring has 35 Tracks per diskette and 18 Sectors per Track. All 35 Tracks are formatted in exactly the same manner and follow the standard IBM format fairly closely. Figure 3 shows the recording format which will be used in this Minifloppy Addendum.

Data which is recorded on the diskette is done so by using frequency encoding. This technique requires that each data bit recorded on the diskette has an associated clock bit recorded with it. Data which is written or read back from the diskette has the form shown in Figure 2.



FIGURE 2
FREQUENCY ENCODED DATA

The encoded bit pattern shown in Figure 2 is binary 101. Refer to Figures 7 and 8 (Page 27) of the Users' Manual and note that the bit cell times shown are exactly 1/2 as long as those shown in this Addendum. Flip-flop U58(A) has been added in series with the Write Clock Signal on the uPD372 (pin 13) in order to divide the clock rate in half, making it compatible with the Minifloppy requirements.

Three special identification marks (Address Marks), are used in the Minifloppy, (four Address Marks are used in the Standard Floppy).

MINIFLOPPY ADDRESS MARKS

ID Address Mark
Data Address Mark
Deleted Data Address Mark

When the Deleted Data Address Mark is written at the beginning of a data field, then the entire contents of the data field will be ignored. Figure 4 shows the recording format for these special codes.



FIGURE 3
MINIFLOPPY RECORDING FORMAT



FIGURE 4
ADDRESS MARKS

DATA CONDITIONER

The uPD372 contains the electronic circuitry for separating the Data and Clock pulses out of the Read Data signal. However, the uPD372 does require some external circuitry so that its internal registers which are clocked on the trailing edge of Ø2 will be presented the proper data. The external circuitry for accomplishing this is called a "Data Conditioner". The floppy disk's Read Signal is asynchronous with the microprocessors' clocks. The Data Conditioner function is to stretch the narrow Read Signal pulses from the floppy so that they will overlap the microprocessor's clock interval, allowing them to be strobed into the uPD372.

The Data Conditioner required for the Minifloppy is much simpler than that required for the Standard floppies. The reason being that the bit cell time is twice as long, namely 8us vs. 4us; thereby eliminating the need for double buffering of the data. The conditions and requirements mentioned on Pages 26, 27 and 28 of the Users' Manual are still applicable. Figure 5 shows the schematic and timing diagram for a Minifloppy Data Conditioner.

The Read Data signal coming from the Minifloppy consists of a string of pulses representing flux reversals on the diskette. U55A is a retriggerable one-shot which times out after 5.8us, detecting the absence of either a data or clock pulse in the Read Data. The Q output of U55A is used as the Read Data input to the uPD372 (RD). The "AND" gate (74Ø8) and the associated RC network on one of its inputs, detects negative going transitions at the output of the RD one-shot. The pulses at point A indicate missing clock or data information during each data cell. These pulses are "OR'd" with the Read Data (7432) producing a pulse stream which contains two pulses per data cell, point B. U55B converts this pulse stream into pulses whose widths are uniform (750ns wide), which may then be sent to the Read Clock input of the uPD372 (RCK).

It should be noted that two (2) RCK pulses are always generated per data cell, even when clock or data pulses are missing. The Ø2 clock pulses (which drive both the microprocessor and uPD372) occur at a 500ns rate, thereby, dictating that the RCK pulses should be 750ns wide for an optimum sampling rate of once per Ø2 pulse. Once a positive transition of RCK is sensed by Ø2, the following Ø2 pulse, clocks the logic level of RD into the uPD372's internal shift register. When the user implements either this or his own Data Conditioner circuit, it is important that he follow the timing constraints outlined in this section of the manual.



FIGURE 5
MINI-FLOPPY DATA CONDITIONER

SAMPLE CONTROLLER DESIGN

Hardware

A Controller design is shown in Figure 6. The Controller is architectured in exactly the same manner as shown in the Users' Manual (reference Pages 31-39). The connections to the Minifloppies are done through connectors J1 and J2 on the right-hand side of the schematic. This design handles only two drives. However, four drives could be accommodated by simply decoding the UAØ and UA1 as well as the UBØ and UB1 lines into two sets of four lines. The figure below shows how this may be achieved.



Figure 7 -- Multiple Drive Decoding

The program for this Controller is contained in a single 1K x 8 bit electrically erasable PROM (Part Number uPD458). However, once the user is confident of his program, a custom ROM (uPD23Ø8) could be made, reducing the parts' costs. The ROM contains the Drives' Reading, Writing, Formatting and Disk Handling Routines.



NOTES

1. THE HOST PROCESSOR AND CONTROLLER PROCESSOR ARE BOTH uPD8080A'S. HOST PROCESSOR SIGNALS ARE PREFIXED WITH H-. CONTROLLER PROCESSOR SIGNALS ARE PREFIXED WITH C-.
2. THE HOST HAS DIRECT MEMORY ACCESS INTO THE CONTROLLER-NOT VICE VERSA.
3. HOST ADDRESS SIGNALS DO NOT REACH THE CONTROLLER MEMORIES UNTIL C-HLDA IS GRANTED DURING A HOST ACCESS. THE HOST IS THEN IN A TW STATE. ONE SHOT U26A KEEPS THE H-READY LINE LOW (TO MAINTAIN ADDRESS AND DATA PATHS) FOR AT LEAST THE 450ns MEMORY ACCESS TIME.
4. H-WR IS ALREADY PRESENT DURING A WRITE CYCLE WHEN THE HOST ADDRESS SIGNALS REACH THE CONTROLLER MEMORIES (HOST IS IN A TW STATE AT C-HLDA). THIS DELAY CIRCUIT KEEPS THE CONTROLLER RAM R/W SIGNAL HIGH FOR THE REQUIRED T_{AW} TIME. SEE uPD2101AL SPECIFICATION.
5. THIS CIRCUIT JAMS AN "FB" (HEX) CODE (ENABLE INTERRUPT) ONTO THE DATA BUS DURING C-INTA.
6. RSØ-RS1 MUST BE STABLE FROM 150ns BEFORE TO 10ns AFTER THE TRAILING EDGE OF Ø1.(SEE uPD372 SPECIFICATIONS) F/F U32 SYNCRONIZES RSØ,RS1 CHANGES WITH THE LEADING EDGE OF Ø2.
7. UNLESS OTHERWISE SPECIFIED ALL RESISTORS 1/4W,±5%

CHIP LIST		
QTY	PART NO.
1	uPD372D
2	uPD2101AL-4
1	uPD2308
1	7400
2	7404
1	7408
1	7416
1	7430
2	7432
4	7474
2	74123
3	74125
1	74157
4	74365
1	uPD8080A
2	uPB8216
1	uPB8224
1	uPB8238
3	74368
1	RESISTOR NETWORK SPRAGUE NO. 914C131XSTR
35	TOTAL

FIGURE 6-uPD372D DUAL MINI-FLOPPY DISC CONTROLLER

3-22-77

Software

The program listing at the end of this Addendum is intended to be a superset of the program in the Users' Manual. The source program contains all instructions necessary for both the Minifloppy and the Standard floppy, all the instructions for both Read/Write and Formatting, and all the instructions for both Single and Multi unit systems. For any specific case, only selected portions of the source program should be assembled. This is accomplished by setting the symbols

```
STD, MINI; RDWR, FMT; MU, NU
```

to the appropriate values. All but the symbol NU are set to either a true or a false value. A true value is defined as 16 one bits which is FFFF in hexadecimal. A false value is defined as 16 zero bits, or 0000 hexadecimal. The last symbol, NU, is set to the number of units (drives) in the system.

The attached assembly listing is assembled for a Minifloppy, with Read/Write and Format routines, for a Multi-Unit system with two units. Therefore, the symbolic values are as follows:

```
 STD EQU FALSE   (Standard = False)
MINI EQU TRUE    (Minifloppy = True)

RDWR EQU TRUE    (Read/Write = True)
 FMT EQU TRUE    (Format = True)

  MU EQU TRUE    (Multi-Unit = True)
  NU EQU   2     (No. Units = 2)
```

Note: For more than 2 units, both hardware and software require some minor modifications.

The memory map for the controller program is as follows:



Figure 8 -- Memory Map

The logic flow for the controller program is simple and is shown as follows:



Figure 9 -- Program Logic Flow

In the attached assembly listing, the source that was used produces assembled code (numbers) in the two left-hand columns. Those portions of the source that were skipped because of the values of MINI, STD, etc., produce no code in the two left-hand columns. Therefore, when reading through the assembly listing, if the two left-hand columns are blank, ignore the source statements to the right. After the basic program is understood, these skipped portions will show the differences between the Mini and the Standard, and the differences between a Single and Multi-Unit system.

```
                ; NEC FLOPPY DISK DRIVE CONTROLLER PROGRAM 04-05-77 1700 (GCY)
                ; CALLING SEQUENCE
                ;       IN ANY SEQUENCE SET THE VALUES FOR THE PARAMETERS:
                ;               UNIT (0 TO (NU-1) )
                ;               TRACK (0 TO (NTRKS-1) )
                ;               SECTR (1 TO NSCTR)
                ;               SCTSZ (4 TO NBSCT) - OPTIONAL: DEFAULT=128
                ;       THE LAST THING TO DO IS TO SET THE COMMAND.
                ;               CMND    =1 FOR READ
                ;                       =2 FOR WRITE
                ;                       =3 FOR SEEK
                ;                       =4 FOR INIT (INITIALIZE)
                ;                       =5 FOR FRMAT (FORMAT)
                ;                       =6 FOR RST0 (RESET)
                ;       THE CONTROLLER SIGNALS COMPLETION BY ZEROING THE
                ;       COMMAND LOCATION.  THEREFORE WAIT FOR CMND=0.
                ;
                ; ASSEMBLY SWITCHES
0000            FALSE   EQU  0
FFFF            TRUE    EQU NOT FALSE
0000            STD     EQU FALSE ;STANDARD FLOPPY
FFFF            MINI    EQU TRUE ;MINIFLOPPY
FFFF            RDWR    EQU TRUE ;READ/WRITE SWITCH
FFFF            FMT     EQU TRUE ;FORMAT SWITCH
FFFF            MU      EQU TRUE ;MULTIPLE UNITS
0002            NU      EQU 2   ;NUMBER OF FLOPPY DISK DRIVE UNITS
                ;
                ; ***** EQUATES FOR USE WITH UPD372 *****
                ;
0000            W0      EQU 0   ;WRITE REGISTER ZERO
0080            W0RST   EQU 80H ;RESET
0040            W0MBL   EQU 40H ;MUST BE LOW
0008            W0HLD   EQU 08H ;HEAD LOAD(NOT USED FOR MINI)
0004            W0LCT   EQU 04H ;LOW CURRENT(MOTOR ON FOR MINI #2)
0002            W0WFR   EQU 02H ;WRITE FAULT RESET(MOTOR ON FOR MINI #1)
                ;
0001            W1      EQU 1   ;WRITE REGISTER ONE
0080            W1CBS   EQU 80H ;CLOCK BIT STROBE
0038            W1CBN   EQU 38H ;CLOCK BITS FOR NORMAL DATA
0010            W1CBI   EQU 10H ;CLOCK BITS FOR INDEX ADDRESS MARK
0000            W1CBD   EQU 00H ;CLOCK BITS FOR ID, DATA,
                                ;OR DELETED DATA ADDRESS MARK
0004            W1UAS   EQU 04H ;UNIT A STROBE
0003            W1UAA   EQU 03H ;UNIT A ADDRESS MASK
                ;
0002            W2      EQU 2   ;WRITE DATA REGISTER
                ;
0003            W3      EQU 3   ;WRITE REGISTER THREE
0080            W3RCS   EQU 80H ;READ CLOCK SET
0040            W3WCS   EQU 40H ;WRITE CLOCK SET
0020            W3STT   EQU 20H ;START READ/WRITE OPERATION
0010            W3WES   EQU 10H ;WRITE ENABLE SET
0008            W3IXS   EQU 08H ;INDEX START
```

```
0004            W3WER   EQU 04H ;WRITE ENABLE RESET
0002            W3CCG   EQU 02H ;CYCLIC CHECK GENERATE
0001            W3CCW   EQU 01H ;READ/WRITE CYCLIC CHECK WORDS
                ;
0004            W4      EQU 4   ;WRITE REGISTER FOUR
0080            W4STS   EQU 80H ;STEP STROBE
0040            W4SID   EQU 40H ;STEP IN OR DIRECTION
0020            W4SOS   EQU 20H ;STEP OUT OR STEP
0004            W4UBS   EQU 04H ;UNIT B STROBE
0003            W4UBA   EQU 03H ;UNIT B ADDRESS MASK
                ;
0006            W6      EQU 6   ;WRITE REGISTER SIX
0004            W6TRR   EQU 04H ;TIMER REQUEST RESET
0002            W6IRR   EQU 02H ;INDEX REQUEST RESET
0001            W6DRR   EQU 01H ;DATA REQUEST RESET(DONE BY HARDWARE)
                ;
0000            R0      EQU 0   ;READ REGISTER ZERO
0080            R0ALH   EQU 80H ;ALWAYS HIGH
0040            R0RYB   EQU 40H ;READY B(NOT USED WITH MINI)
0030            R0UBA   EQU 30H ;UNIT B ADDRESS MASK
0008            R0ERR   EQU 08H ;ERROR
0004            R0TRQ   EQU 04H ;TIMER REQUEST
                                ;(NO INTERRUPT GENERATED)
0002            R0IRQ   EQU 02H ;INDEX REQUEST
                                ;(INTERRUPT CLEARED BY SOFTWARE)
0001            R0DRQ   EQU 01H ;DATA REQUEST
                                ;(INTERRUPT CLEARED BY HARDWARE)
                ;
0001            R1      EQU 1   ;READ REGISTER ONE
0080            R1WRT   EQU 80H ;WRITE MODE
0040            R1T00   EQU 40H ;TRACK 00
0020            R1DER   EQU 20H ;DATA ERROR (CRC)
0010            R1COR   EQU 10H ;COMMAND OVERRUN
0008            R1RYA   EQU 08H ;READY A(NOT USED WITH MINI)
0004            R1WFT   EQU 04H ;WRITE FAULT(NOT USED WITH MINI)
0003            R1UAA   EQU 03H ;UNIT A ADDRESS MASK
                ;
0002            R2      EQU 2   ;READ DATA REGISTER
                ;
                ;
0003            NTRYS   EQU 3   ;NUMBER OF READ RETRYS
                        IF MINI
0023            NTRKS   EQU 35  ;NO. OF TRACKS/DISK
0012            NSCTR   EQU 18  ;NO. OF SECTORS/TRACK
0080            NBSCT   EQU 128 ;NO. OF BYTES/SECTOR
07D0            TMLIM   EQU 2000 ;IDLE TIME LIMIT IN MS
0028            TTACS   EQU 40  ;TRACK TO TRACK ACCESS IN MS
000A            STLNG   EQU 10  ;HEAD SETTLING TIME IN MS
0008            RSKIP   EQU 8   ;NO. OF GAP BYTES TO SKIP IN READ
0006            WSKIP   EQU 6   ;NO. OF GAP BYTES TO SKIP IN WRITE
000F            NFPOI   EQU 16-1 ;NO. OF FF'S IN POST INDEX GAP
0006            NFGP2   EQU 6   ;NO. OF FF'S IN GAP 2
0011            NFGP3   EQU 17  ;NO. OF FF'S IN GAP 3
                        ENDIF
```

```
                        IF STD
                NTRKS   EQU 77  ;NO. OF TRACKS/DISK
                NSCTR   EQU 26  ;NO. OF SECTORS/TRACK
                NBSCT   EQU 128 ;NO. OF BYTES/SECTOR
                LHCTK   EQU 43  ;LAST HIGH CURRENT TRACK
                TMLIM   EQU 667 ;IDLE TIME LIMIT IN MS
                TTACS   EQU 10  ;TRACK TO TRACK ACCESS IN MS
                RSKIP   EQU 13  ;NO. OF GAP BYTES TO SKIP IN READ
                WSKIP   EQU 11  ;NO. OF GAP BYTES TO SKIP IN WRITE
                NFPRI   EQU 40-1 ;NO. OF FF'S IN PRE INDEX GAP
                NFPOI   EQU 26  ;NO. OF FF'S IN POST INDEX GAP
                NFGP2   EQU 11  ;NO. OF FF'S IN GAP 2
                NFGP3   EQU 27  ;NO. OF FF'S IN GAP 3
                        ENDIF
0480            STACK   EQU 0480H
                ;
                ;
0000                    ORG 0
                ;
                ; ***** RESET *****
0000 F3         RST0:   DI              ;RESET
0001 3E80               MVI A,W0RST
0003 D300               OUT W0          ;RESET 372, CLEAR WRITE ENABLE
0005 318004             LXI SP,STACK
                ;INITIALIZE DATA AREA TO ZERO
0008 210004     RS020:  LXI H,CMND      ;HL=ADR(DATA AREA)
000B 061A               MVI B,NB        ;B=NO. OF BYTES
000D AF                 XRA A
000E 77         RS030:  MOV M,A         ;M=0
000F 23                 INX H
0010 05                 DCR B           ;DONE?
0011 C20E00             JNZ RS030       ;NO
                ; INITIALIZE SECTOR SIZE
0014 3E80               MVI A,NBSCT
0016 320404             STA SCTSZ
                ; INITIALIZE ALL UNITS
                        IF MU
0019 010100             LXI     B,NU-1  ;START WITH UNIT (NU-1)
                        ENDIF
001C CD2600     RS010:  CALL    INIT
                        IF MU
001F 0D                 DCR     C       ;LAST UNIT?
0020 F21C00             JP      RS010   ;NO, GO DO NEXT UNIT
                        ENDIF
0023 C34E00             JMP RT010
                ;
                ;
                ; ***** INITIALIZE DISK UNIT SUBROUTINE *****
                ;  INPUT: BC=UNIT#
                ;  REGISTERS: AF,DE,HL
                ;  STACK PAIRS: 2
0026            INIT    EQU     $
                        IF NOT MU
                        CALL    UASLC
```

```
                        ENDIF
                        IF MU
0026 CDDF02             CALL UBSLC
                        ENDIF
                        IF MINI
0029 CDBB02             CALL    MTOFF   ;TURN MOTOR OFF AND UNLOAD HEAD
                        ENDIF
                        IF STD
                        CALL UNLD       ;UNLOAD HEAD
                        ENDIF
                ; MOVE HEAD TO TRACK ZERO
002C 1E22               MVI     E,NTRKS-1 ;E=LOOP LIMIT
002E DB01       IN010:  IN R1           ;READ STATUS
0030 E640               ANI R1T00       ;TRACK 0?
0032 C23C00             JNZ IN020       ;YES, DONE
0035 CD8302             CALL STO        ;NO, STEP OUT
0038 1D                 DCR     E       ;LIMIT REACHED?
0039 C22E00             JNZ     IN010   ;NO, CHECK AGAIN
003C 211204     IN020:  LXI     H,TKPTR
                        IF MU
003F 09                 DAD     B       ;ADD UNIT# TO HL
                        ENDIF
0040 70                 MOV M,B         ;TKPTR=0
0041 C9                 RET
                ;
                ; RETURN FROM COMMAND
0042 F3         RETRN:  DI
0043 CA4B00             JZ RT005        ;WAS THERE AN ERROR?
0046 3E01               MVI A,1         ;YES, SET THE
0048 320504             STA MERF        ;MASTER ERROR FLAG
004B CDD200     RT005:  CALL    TRSET   ;RESET IDLE TIME
004E 3E03       RT010:  MVI     A,NTRYS
0050 321004             STA RRTRY       ;RESET NO. OF READ RETRYS
0053 AF                 XRA A
0054 320004             STA CMND        ;RESET COMMAND TO ZERO
                ;
                ; IDLE LOOP - CHECK FOR A COMMAND
                ;
                ; CHECK IDLE TIME
0057 DB00       IDL10:  IN      R0      ;INPUT CLOCK STATUS
0059 E604               ANI     R0TRQ   ;DID CLOCK TICK?
005B CA8800             JZ      IDL40   ;NO, GO CHECK FOR COMMAND
005E D306               OUT     W6      ;YES, RESET CLOCK, UPDATE IDLE TIME
0060 010100             LXI     B,NU-1  ;BC=UNIT#
0063 211404     IDL20:  LXI     H,ITIME
                        IF MU
0066 CDF702             CALL    DINDX
                        ENDIF
0069 C5                 PUSH    B       ;SAVE BC
006A 01D007             LXI     B,TMLIM ;BC=TIME LIMIT
006D 5E                 MOV     E,M
006E 23                 INX     H
006F 56                 MOV     D,M     ;DE=ELAPSED IDLE TIME
0070 13                 INX     D       ;INCREMENT IDLE TIME
```

```
0071 72                 MOV     M,D
0072 2B                 DCX     H
0073 73                 MOV     M,E     ;UPDATE IDLE TIME
0074 7A                 MOV     A,D     ;TEST MSB
0075 B8                 CMP     B       ;D=B?
0076 C28300             JNZ     IDL30   ;NO, CONTINUE
0079 7B                 MOV     A,E
007A B9                 CMP     C       ;E=C?
007B C28300             JNZ     IDL30   ;NO, CONTINUE
007E C1                 POP     B       ;RESTORE BC
007F C5                 PUSH    B       ;MAINTAIN STACK POSITION
                        IF MINI
0080 CDBB02             CALL    MTOFF   ;YES, TIME LIMIT REACHED,
                                        ;TURN OFF MOTOR, UNLOAD HEAD
                        ENDIF
                        IF STD
                        CALL    UNLD    ;YES, TIME LIMIT REACHED,
                                        ;UNLOAD HEAD
                        ENDIF
0083 C1         IDL30:  POP     B       ;RESTORE STACK POSITION
                        IF MU
0084 0D                 DCR     C       ;LAST UNIT?
0085 F26300             JP      IDL20   ;NO, KEEP CHECKING
                        ENDIF
                ; CHECK FOR COMMAND
0088 3A0004     IDL40:  LDA     CMND
008B B7                 ORA     A       ;IS THERE A COMMAND?
008C CA5700             JZ      IDL10   ;NO, STAY IN IDLE LOOP
                ; EXECUTE COMMAND
                ;  A=COMMAND(1-NCMDS)
008F 4F         EXEC:   MOV C,A         ;SAVE COMMAND IN C
0090 110504             LXI D,MERF      ;ZERO FLAGS
0093 060B               MVI B,NF
0095 AF                 XRA A
0096 12         EX005:  STAX D
0097 13                 INX D
0098 05                 DCR B
0099 C29600             JNZ EX005
009C 3E06               MVI A,NCMDS
009E B9                 CMP C           ;IS CMND OK?
009F D2A800             JNC     EX010   ;YES
00A2 320604             STA CMDER       ;NO, SET FLAG
00A5 C3E700             JMP ERROR
                ; CHECK ALL PARAMETERS
00A8 110104     EX010:  LXI D,UNIT      ;DE=ADR(PARAMETERS)
00AB 21F000             LXI H,LMTBL     ;HL=ADR(LIMIT TABLE)
00AE 0604               MVI B,NP        ;B=NO. OF PARAMETERS
00B0 1A         EX020:  LDAX D          ;A=PARAMETER
00B1 BE                 CMP M           ;LOWER LIMIT OK?
00B2 DAE200             JC EX040        ;NO, ERROR
00B5 23                 INX H           ;YES
00B6 BE                 CMP M           ;UPPER LIMIT OK?
00B7 D2E200             JNC EX040       ;NO, ERROR
00BA 23                 INX H           ;YES
```

```
00BB 13              INX D
00BC 05              DCR B              ;DONE?
00BD C2B000          JNZ EX020          ;NO
                ; COMMAND AND PARAMETERS OK
00C0 79              MOV A,C            ;YES, A=COMMAND
00C1 21F800          LXI H,CTBL
00C4 3D              DCR A              ;A=(0-(N-1))
00C5 07              RLC                ;A=2*A
00C6 5F              MOV E,A
00C7 1600            MVI D,0
00C9 19              DAD D              ;HL=ADR(ADR)
00CA 5E              MOV E,M
00CB 23              INX H
00CC 56              MOV D,M            ;DE=ADR
00CD 214200          LXI H,RETRN
00D0 E5              PUSH H             ;(SP)=RETURN ADDRESS
00D1 D5              PUSH    D          ;SAVE ROUTINE ADDRESS
                ;IDLE TIME RESET
00D2 3A0104     TRSET:  LDA     UNIT
00D5 4F                 MOV     C,A     ;C=UNIT #
00D6 0600               MVI     B,0     ;BC=UNIT#
00D8 211404             LXI     H,ITIME
                        IF MU
00DB CDF702             CALL    DINDX
                        ENDIF
00DE 70                 MOV     M,B     ;SET IDLE TIME TO ZERO
00DF 23                 INX     H
00E0 70                 MOV     M,B
00E1 C9                 RET             ;VECTOR TO ROUTINE, OR
                                        ;  RETURN TO CALLER
                ;
00E2 3E01       EX040:  MVI A,1
00E4 320704             STA PRMER       ;SET PARAMETER FLAG
                ;
00E7 320504     ERROR:  STA MERF        ;SET MASTER ERROR FLAG
00EA 318004             LXI SP,STACK    ;RESET SP
00ED C34E00             JMP RT010
                ;
                ;LIMIT TABLE (UPPER AND LOWER FOR PARAMETERS)
00F0 0002       LMTBL:  DB 0,NU         ;UNIT
00F2 0023               DB 0,NTRKS      ;TRACK
00F4 0113               DB 1,NSCTR+1    ;SECTR
00F6 0481               DB 4,NBSCT+1    ;SCTSZ
                ;
                ; COMMAND TABLE
00F8 8F01       CTBL:   DW READ         ;1
00FA 0C02               DW WRITE        ;2
00FC 0401               DW SEEK         ;3
00FE 2600               DW INIT         ;4
0100 FF02               DW FRMAT        ;5
0102 0000               DW RST0         ;6
0006            NCMDS   EQU ($-CTBL)/2
                ;
                ;
```

```
                ; ***** SEEK TRACK ROUTINE *****
                ; INPUT: BC=UNIT#
                ;  REGISTERS: AF,D,HL
                ;  SUBROUTINES: UASLC,UBSLC,STI,STO
                ;
0104            SEEK    EQU     $
                        IF MU
0104 CDDF02             CALL    UBSLC
                        ENDIF
                        IF NOT MU
                        CALL    UASLC
                        ENDIF
0107 211204             LXI     H,TKPTR
                        IF MU
010A 09                 DAD     B       ;ADD UNIT# TO HL
                        ENDIF
010B 3A0204             LDA TRACK       ;A=TRACK DESIRED
010E BE                 CMP M
010F C8                 RZ              ;TRACK=TKPTR
                        IF MINI
0110 E5                 PUSH    H       ;SAVE ADR(TKPTR)
                        ENDIF
0111 DA1A01             JC      SK010   ;TKPTR>TRACK
0114 CD7C02             CALL STI        ;TKPTR<TRACK
                        IF STD
                        JMP SEEK
                        ENDIF
                        IF MINI
0117 C31D01             JMP     SK020
                        ENDIF
011A CD8302     SK010:  CALL STO        ;TKPTR>TRACK
                        IF STD
                        JMP SEEK
                        ENDIF
                        IF MINI
011D E1         SK020:  POP     H       ;HL=ADR(TKPTR)
011E 3A0204             LDA     TRACK
0121 BE                 CMP     M       ;TRACK=TKPTR?
0122 C20401             JNZ     SEEK    ;NO, KEEP SEEKING
0125 060A               MVI     B,STLNG ;YES, ALLOW HEAD TIME TO SETTLE
0127 CDE602             CALL    DELAY
012A C9                 RET
                        ENDIF
                        IF RDWR
                ;
                ; READ ID RECORD ROUTINE
                ;
                ;
                ;REGISTERS: A,F,B,C,DE,HL
                ;
012B CD0401     RID:    CALL SEEK       ;POSITION HEAD
                        IF MU
012E CDD802             CALL    UASLC   ;SELECT UNIT WITH A LINES
                        ENDIF
```

```
                        IF MINI
0131 CD9E02             CALL    MTRON   ;TURN ON MOTOR AND LOAD HEAD
                        ENDIF
                        IF STD
                        CALL    RDYA    ;CHECK FOR UNIT READY
                        CALL HDLD       ;LOAD HEAD
                        ENDIF
                ;
0134 0E04               MVI     C,4     ;STORE LIMIT OF REVOLUTIONS OF
                                        ;DISK WITHOUT FINDING CORRECT ID
                                        ;RECORD.  USE 4 TO GUARANTEE
                                        ;THREE COMPLETE REVOLUTIONS.
                ;
0136 210204     RIA:    LXI H,TRACK     ;INITIALIZE TRACK/SECTR POINTER
0139 110A04             LXI D,WTRK      ;INITIALIZE FLAG POINTER
013C AF                 XRA     A
013D 47                 MOV B,A         ;SET B=0
013E D303               OUT W3          ;RESET STT (FOR RETRY)
                                        ;
0140 3EA0               MVI A,W3RCS+W3STT ;RCS=1, STT=1
0142 D303               OUT     W3      ;GO TO READ CLOCK.  SET STT TO AUTO-
                                        ;MATICALLY START READ OPERATION WHEN
                                        ;ADDRESS MARK IS READ.
                                        ;
0144 FB                 EI              ;ENABLE INTERRUPT AND WAIT FOR
0145 76                 HLT             ;ADDRESS MARK TO BE READ.
                ;
                ;INTERRUPT(ADDRESS MARK)
                ;
                ;       (EI)
0146 DB02               IN      R2      ;READ DATA
0148 EEFE               XRI     0FEH    ;IS IT AN I.D. ADDRESS MARK?
014A C27801             JNZ     RIM     ;NO: JUMP TO RIM
014D 76                 HLT             ;YES: WAIT FOR NEXT INTERRUPT.
                ;
                ;INTERRUPT(TRACK ADDRESS)
                ;
                ;       (EI)
014E DB02               IN      R2      ;READ TRACK ADDRESS BYTE.
0150 AE                 XRA M           ;COMPARE WITH DESIRED TRACK
0151 12                 STAX D          ;WTRK =0 FOR OK, =NON-ZERO FOR ERROR
0152 13                 INX D           ;DE POINTS TO NEXT FLAG
0153 23                 INX     H       ;HL POINTS TO SECTR
0154 76                 HLT             ;WAIT FOR NEXT INTERRUPT
                        IF STD
                ;
                ;INTERRUPT(FIRST ZERO BYTE)
                ;
                ;       (EI)
                        IN      R2      ;READ ZERO BYTE
                        STAX D          ;ZERO1 =0 FOR OK, =NON-ZERO FOR ERROR
                        INX D           ;DE POINTS TO ZERO2
                                        ;
                        HLT             ;WAIT FOR NEXT INTERRUPT
```

```
                        ENDIF
                    ;
                    ;INTERRUPT(SECTOR ADDRESS)
                    ;
                    ;       (EI)
0155 DB02                   IN      R2          ;READ SECTOR ADDRESS BYTE
0157 AE                     XRA M               ;COMPARE WITH DESIRED SECTOR
0158 47                     MOV B,A             ;B =0 FOR OK, =NON-ZERO FOR ERROR
0159 3E21                   MVI A,W3STT+W3CCW
015B D303                   OUT W3              ;SEND COMMAND TO W3.
                                                ;THIS COMMAND SETS CCW.  (STT
                                                ;BIT MUST ALSO BE A ONE TO AVOID
                                                ;RESETTING STT.)  THE BIT RING PULSE
                                                ;(BRP) FOLLOWING THE SETTING OF CCW
                                                ;WILL START A BIT BY BIT COMPARISON
                                                ;OF THE DATA READ FROM THE DISK WITH
                                                ;THE DATA READ FROM THE CRC REGISTER.
                                                ;(ALTHOUGH THE CPU WILL HAVE READ
                                                ;A COMPLETED BYTE AT THE NEXT
                                                ;BRP, THE DISK DRIVE HEAD WILL BEGIN
                                                ;READING THE 1ST CRC BYTE.)
                                                ;
015D 76                     HLT                 ;WAIT FOR NEXT INTERRUPT.
                    ;
                            IF STD
                    ;INTERRUPT(SECOND ZERO BYTE)
                    ;
                    ;       (EI)
                            IN      R2          ;READ 2ND ZERO BYTE
                                                ;
                            STAX D              ;ZERO2 =0 FOR OK, =NON-ZERO FOR ERROR
                            INX D               ;DE POINTS TO CRCID
                                                ;
                            HLT                 ;WAIT FOR NEXT INTERRUPT.
                            ENDIF
                    ;
                    ;INTERRUPT(CRC BYTE 1)
                    ;
                    ;       (EI)
015E 3E20                   MVI A,W3STT         ;TURN OFF CCW
                                                ;
0160 D303                   OUT     W3          ;SEND COMMAND TO W3
                                                ;STT=1, CCW=0.  CCW IS RESET.
                                                ;AT NEXT BRP BIT-BY-BIT CRC
                                                ;COMPARISON WILL END.
                                                ;
0162 76                     HLT                 ;WAIT FOR NEXT INTERRUPT.
                    ;
                    ;INTERRUPT(CRC BYTE 2)
                    ;
                    ;       (EI)
0163 DB01                   IN      R1          ;INTERRUPT CAUSED BY 2ND CRC BYTE
                                                ;
0165 E620                   ANI R1DER           ;WAS THERE A CRC ERROR?
```

```
0167 12              STAX D          ;CRCID =Ø FOR OK, =NON-ZERO FOR ERROR
0168 78              MOV A,B
0169 B7              ORA A           ;SECTOR OK?
016A C278Ø1          JNZ RIM         ;NO, TRY AGAIN
016D 76              HLT             ;YES
                ;
                ;INTERRUPT (FIRST GAP BYTE)
                ;
                ;       (EI)
016E EB              XCHG            ;HL POINTS TO CRCID, A=Ø
016F B6              ORA M           ;TEST CRCID
                     IF STD
                     DCX H
                     ORA M           ;TEST ZERO2
                     DCX H
                     ORA M           ;TEST ZERO1
                     ENDIF
0170 76              HLT
                ;
                ;INTERRUPT (2ND GAP BYTE)
                ;
                ;       (EI)
0171 2B              DCX H
0172 B6              ORA M           ;TEST WTRK. (IS TRACK ADDRESS
                                     ;READ EQUAL TO SOFTWARE TRACK
                                     ;POINTER?)
0173 C278Ø1          JNZ RIM         ;ONE OF THE ABOVE IN ERROR, TRY AGAIN.
0176 76              HLT
                ;
                ;INTERRUPT (3RD GAP BYTE)
                ;
                ;       (EI)
0177 C9              RET             ;NORMAL RETURN, ZERO FLAG=1
                ;
                ;ERROR
                ;
0178 DBØØ       RIM:    IN      RØ      ;READ STATUS
017A E602               ANI RØIRQ       ;WAS INTERRUPT AN INDEX REQUEST?
017C CA36Ø1             JZ      RIA     ;NO: WAIT FOR NEXT MARK
017F D3Ø6               OUT W6          ;YES, IRQ RESET
0181 ØD                 DCR C           ;DECREMENT LIMIT
0182 C236Ø1             JNZ     RIA     ;WAIT FOR NEXT MARK IF NOT 3RD COMPLETE
                                        ;REVOLUTION OF DISK WITHQUT SUCCESS
                                        ;
0185 AF                 XRA A           ;QUIT
0186 D3Ø3               OUT     W3      ;RESET STT
                                        ;
                                        ;
0188 3EØ1               MVI A,1
018A 32Ø9Ø4             STA NOGO        ;COULD NOT FIND REQUESTED ID
018D B7                 ORA A           ;CLEAR ZERO FLAG
018E C9                 RET             ;ERROR RETURN
                ;
                ; ***** READ DATA RECORD ROUTINE *****
```

```
                ;
                ;REGISTERS: A,F,B,C,DE,HL
                ;
                ;  CALL READ ID RECORD FIRST
                ;
018F CD2B01     READ:   CALL RID          ;READ ID
0192 C0                 RNZ               ;ERROR IN RID, RETURN
                ;(3 GAP BYTES HAVE BEEN READ, HEAD IS READING 4TH)
0193 3E60               MVI A,W3WCS+W3STT
0195 D303               OUT W3            ;SET WRITE CLOCK, LEAVE STT SET
0197 0604               MVI B,RSKIP-4     ;PASS GAP BYTES 4 THRU (RSKIP-1)
0199 76         RGAP:   HLT
                ;
                ;INTERRUPT (GAP BYTE 4 THRU (RSKIP-1))
                ;
                ;       (EI)
019A 05                 DCR     B
019B C29901             JNZ     RGAP
019E 76                 HLT               ;WAIT FOR GAP BYTE #RSKIP.  HEAD HAS
                                          ;NOW PASSED AREA IN GAP THAT CONTAINS
                                          ;UNKNOWN INFORMATION GENERATED WHEN
                                          ;WRITE CURRENT WAS TURNED ON TO WRITE
                                          ;DATA RECORD.
                ;
                ;INTERRUPT (GAP BYTE #RSKIP)
                ;
                ;       (EI)
019F 3E40               MVI A,W3WCS       ;RESET STT, SET WRITE CLOCK TO
01A1 D303               OUT W3            ;PREVENT INTERRUPTS UNTIL FOLLOWING
                                          ;IS DONE.
                                          ;
01A3 218004             LXI H,BUFFR       ;SET HL TO 1ST ADDRESS OF
                                          ;STORAGE BUFFER.
                                          ;
01A6 1621               MVI D,W3STT+W3CCW ;STORE COMMAND TO SET CCW IN
                                          ;D REGISTER.
                                          ;
01A8 0EFB               MVI     C,0FBH    ;STORE DATA ADDRESS MARK CODE
                                          ;IN C.
                                          ;
01AA 3A0404             LDA SCTSZ         ;SET SECTOR SIZE
01AD D603               SUI 3
01AF 47                 MOV B,A           ;SAVE COUNT IN B
                                          ;
01B0 3EA0               MVI A,W3RCS+W3STT ;SET READ CLOCK, SET STT.
01B2 D303               OUT W3
01B4 76                 HLT               ;WAIT FOR ADDRESS MARK.
                ;
                ;INTERRUPT (ADDRESS MARK)
                ;
                ;       (EI)
01B5 DB02               IN R2             ;READ BYTE
01B7 B9                 CMP C             ;IS IT A DATA ADDRESS MARK?
01B8 C2EB01             JNZ     MARK      ;NO: JUMP TO MARK
```

```
01BB 76                 HLT             ;WAIT FOR FIRST DATA BYTE
                ;
                ;INTERRUPT (DATA BYTE 1)
                ;
                ;       (EI)
01BC DB02               IN R2           ;YES: READ FIRST DATA BYTE
01BE 77                 MOV     M,A     ;STORE FIRST DATA BYTE
                                        ;
                ; READ LOOP
01BF 23         RLOOP:  INX     H       ;INCREMENT BUFFER POINTER
01C0 76                 HLT
                ;
                ;INTERRUPT (DATA BYTES 2 THRU (SCTSZ-2) )
                ;
                ;       (EI)
01C1 DB02               IN R2
01C3 77                 MOV     M,A     ;STORE DATA BYTE IN BUFFER
01C4 05                 DCR     B       ;DONE?
01C5 C2BF01             JNZ     RLOOP   ;NO, READ NEXT BYTE
01C8 23                 INX     H       ;INCREMENT BUFFER POINTER
01C9 76                 HLT
                ;
                ;INTERRUPT (DATA BYTE #(SCTSZ-1) )
                ;
                ;       (EI)
01CA DB02               IN R2           ;READ AND STORE NEXT TO
                                        ;LAST DATA BYTE
01CC 77                 MOV     M,A
                                        ;
01CD 7A                 MOV     A,D     ;SET CCW
01CE D303               OUT W3
01D0 76                 HLT
                ;
                ;INTERRUPT (DATA BYTE #SCTSZ)
                ;
                ;       (EI)
01D1 23                 INX H           ;READ AND STORE LAST DATA BYTE
01D2 DB02               IN R2
01D4 77                 MOV     M,A
01D5 76                 HLT
                ;
                ;INTERRUPT (FIRST CRC BYTE)
                ;
                ;       (EI)
01D6 3E20               MVI A,W3STT
01D8 D303               OUT W3          ;RESET CCW
01DA 76                 HLT
                ;
                ;INTERRUPT (2ND CRC BYTE)
                ;
                ;       (EI)
01DB DB01               IN R1           ;READ STATUS
01DD 47                 MOV B,A         ;SAVE STATUS
01DE AF                 XRA A
```

```
01DF D303              OUT W3          ;RESET STT.  (372 GOES TO WRITE
                                       ;CLOCK AUTOMATICALLY.)
                                       ;
01E1 78                MOV A,B         ;RECALL STATUS
                                       ;
01E2 E620              ANI R1DER       ;IS THERE A CRC ERROR?
01E4 320C04            STA CRCDR       ;SET CRC DATA RECORD FLAG
01E7 C8                RZ              ;NO, NORMAL RETURN
                                       ;
                                       ;
                                       ;READ RECORD BUT FOUND
01E8 C30002            JMP MK030       ;CRC ERROR
                                       ;
01EB AF        MARK:   XRA     A       ;RESET STT
01EC D303              OUT W3
01EE DB02              IN R2           ;READ MARK AGAIN
01F0 D6F8              SUI 0F8H        ;IS IT A "DELETED DATA MARK"?
01F2 320D04            STA ILLMK       ;SET ILLEGAL MARK FLAG
01F5 C2FC01            JNZ MK010       ;ILLEGAL MARK
01F8 3C                INR A           ;DELETED DATA MARK
01F9 C3FD01            JMP MK020
01FC AF        MK010:  XRA A           ;ILLEGAL MARK
01FD 320E04    MK020:  STA DELMK       ;SET DELETED DATA MARK FLAG
0200 211004    MK030:  LXI H,RRTRY     ;CHECK FOR RETRY
0203 35                DCR M
0204 C28F01            JNZ READ        ;TRY AGAIN
0207 AF                XRA A
0208 D303              OUT W3          ;RESET STT
020A 3C                INR A           ;CLEAR ZERO FLAG TO
                                       ;INDICATE AN ERROR CONDITION
020B C9                RET
               ;
               ; ***** WRITE DATA RECORD ROUTINE *****
               ;
               ;REGISTERS: A,F,B,C,DE,HL
               ;
               ;  CALL READ ID RECORD ROUTINE FIRST
               ;
020C CD2B01    WRITE:  CALL RID        ;READ ID
020F C0                RNZ             ;ERROR IN RID, RETURN
               ;(3 GAP BYTES HAVE BEEN READ, HEAD IS READING 4TH)
0210 0601              MVI B,WSKIP-5   ;COUNT INTERRUPTS FROM ID
0212 76        WGAP:   HLT             ;RECORD.  (HEAD WILL THEN BE
               ;
               ;INTERRUPT
               ;
               ;       (EI)
0213 05                DCR     B       ;READING (WSKIP-1) BYTE)
0214 C21202            JNZ     WGAP
                                       ;
0217 3EB8              MVI A,W1CBS+W1CBN ;SET CLOCK BITS AND STROBE
0219 D301              OUT W1          ;SET WRITE CLOCK LOGIC TO WRITE
                                       ;ALL CLOCK BITS ("FF" CLOCK BITS)
                                       ;FOR DATA
```

```
                                        ;
021B AF                 XRA     A       ;SET WRITE DATA REGISTER TO ØØ.
021C D302               OUT W2
021E 76                 HLT             ;WAIT FOR INTERRUPT
                ;
                ;INTERRUPT
                ;
                ;       (EI)
                                        ;HEAD IS READING GAP BYTE #WSKIP.
                                        ;
021F 3E70               MVI A,W3WCS+W3STT+W3WES ;WCS, STT, WES = 1
0221 D303               OUT W3          ;WRITE CURRENT AND WRITE CLOCK
                                        ;WILL START AT NEXT BRP
0223 DB01               IN R1           ;READ STATUS
0225 E604               ANI R1WFT       ;WRITE FAULT?
0227 CA3202             JZ WR010        ;NO, CONTINUE
                        IF STD
                        MVI A,W0WFR
                        CALL SETW0      ;WRITE FAULT RESET
                        MVI A,W0WFR
                        CALL CLRW0      ;CLEAR RESET BIT
                        ENDIF
022A 3E01               MVI A,1
022C 320F04             STA WRITF       ;YES, SET WRITE FAULT FLAG
022F C3E700             JMP ERROR
                                        ;
0232 76         WR010:  HLT             ;WAIT FOR INTERRUPT
                ;
                ;INTERRUPT
                ;
                ;       (EI)            ;LAST FF GAP BYTE READ, 372
                                        ;SWITCHES TO WRITE MODE.
0233 76                 HLT             ;HEAD BEGINS WRITING A ØØ
                                        ;IN GAP BYTE #(WSKIP+1).
                        IF STD
                ;
                ;INTERRUPT
                ;
                ;       (EI)
                        HLT             ;HEAD STARTS WRITING ØØ
                                        ;IN BYTE #(NO. OF GAP BYTES - 4).
                ;
                ;INTERRUPT
                ;
                ;       (EI)
                        HLT             ;HEAD STARTS WRITING ØØ
                                        ;IN BYTE #(NO. OF GAP BYTES - 3).
                        ENDIF
                ;
                ;INTERRUPT
                ;
                ;       (EI)
0234 218004             LXI H,BUFFR     ;SET H,L TO START OF WRITE BUFFER
0237 76                 HLT             ;HEAD STARTS WRITING ØØ IN
```

```
                                        ;BYTE #(NO. OF GAP BYTES - 2).
                ;
                ;INTERRUPT
                ;
                ;       (EI)
0238 06FB               MVI     B,0FBH  ;LOAD DATA MARK IN B
                                        ;HEAD STARTS WRITING 00 IN
                                        ;NEXT TO LAST GAP BYTE.
                                        ;
023A 0E22               MVI C,W3STT+W3CCG ;STORE SET CCG COMMAND IN C
                                        ;
023C 16B8               MVI D,W1CBS+W1CBN ;STORE "FF" CLOCK PATTERN
                                        ;COMMAND IN D
                                        ;
023E 1E20               MVI E,W3STT     ;STORE RESET CCG COMMAND IN E
                                        ;
0240 3E80               MVI A,W1CBS+W1CBD ;STORE "C7" DATA MARK CLOCK
                                        ;PATTERN COMMAND IN A
                                        ;
0242 76                 HLT             ;WAIT FOR INTERRUPT
                ;
                ;INTERRUPT
                ;
                ;       (EI)
0243 D301               OUT W1          ;SET "C7" DATA MARK CLOCK PATTERN.
                                        ;HEAD STARTS WRITING 00 IN
                                        ;LAST GAP BYTE.
                                        ;
0245 78                 MOV     A,B     ;SET "FB" DATA BITS FOR
0246 D302               OUT W2          ;DATA MARK
                                        ;
0248 79                 MOV     A,C     ;SET CCG.  THIS CAUSES CRC
0249 D303               OUT W3          ;CALCULATION TO BEGIN AT NEXT BRP.
                                        ;
024B 7A                 MOV     A,D     ;GET "FF" DATA BIT CLOCK
                                        ;PATTERN IN A
                                        ;
024C 76                 HLT             ;WAIT FOR INTERRUPT.
                ;
                ;INTERRUPT
                ;
                ;       (EI)
024D D301               OUT W1          ;SET "FF" DATA BIT CLOCK PATTERN
                                        ;FOR NEXT BYTE.  HEAD NOW BEGINS
                                        ;WRITING DATA MARK
                                        ;
024F 7B                 MOV     A,E     ;RESET CCG.  (CCG MUST BE RESET
0250 D303               OUT W3          ;BEFORE NEXT BRP OR CRC CALCULATION
                                        ;WOULD BEGIN AGAIN.)
                                        ;
0252 7E                 MOV     A,M     ;LOAD FIRST DATA BYTE IN
0253 D302               OUT W2
                                        ;
0255 76                 HLT             ;WAIT FOR INTERRUPT
```

```
                ;
                ;INTERRUPT
                ;
                ;       (EI)            ;DATA MARK WRITTEN, HEAD
                                        ;STARTS WRITING DATA BYTE #1.
0256 3A0404             LDA SCTSZ       ;SET SECTOR SIZE
0259 3D                 DCR A
025A 47                 MOV B,A         ;SAVE COUNT IN B.  HEAD
                                        ;BEGINS WRITING FIRST DATA BYTE.
                                        ;
                ; WRITE LOOP
025B 23         WLOOP:  INX     H       ;WRITE DATA BYTES 2 THRU NBSCT
025C 7E                 MOV     A,M
025D D302               OUT W2
025F 76                 HLT
                ;
                ;INTERRUPT
                ;
                ;       (EI)
0260 05                 DCR     B
0261 C25B02             JNZ     WLOOP
                                        ;
0264 3E21               MVI A,W3STT+W3CCW ;SET CCW.  IN WRITE MODE THE 372
0266 D303               OUT W3          ;WILL BEGIN WRITING BITS FROM THE
                                        ;CRC REGISTER AT THE NEXT BRP
                                        ;FOLLOWING THE SETTING OF CCW.
                                        ;(HEAD IS WRITING LAST DATA BYTE)
                                        ;
0268 76                 HLT             ;WAIT FOR INTERRUPT
                ;
                ;INTERRUPT
                ;
                ;       (EI)            ;LAST DATA BYTE WRITTEN.
0269 76                 HLT             ;HEAD STARTS WRITING FIRST CRC BYTE
                ;
                ;INTERRUPT
                ;
                ;       (EI)            ;FIRST CRC BYTE WRITTEN.
026A 3EFF               MVI     A,0FFH  ;LOAD FF GAP BYTE IN WRITE DATA
026C D302               OUT W2          ;REGISTER (HEAD BEGINS WRITING
                                        ;2ND CRC BYTE.)
                                        ;
026E 3E20               MVI A,W3STT     ;RESET CCW COMMAND.  CRC BIT
0270 D303               OUT W3          ;WRITING WILL STOP AT NEXT BRP.
                                        ;
0272 76                 HLT             ;WAIT FOR INTERRUPT.
                ;
                ;INTERRUPT
                ;
                ;       (EI)            ;2ND CRC BYTE WRITTEN, HEAD
                                        ;STARTS WRITING FF GAP BYTE.
0273 3E24               MVI A,W3STT+W3WER ;WRITE ENABLE RESET.  WRITE
0275 D303               OUT W3          ;CURRENT WILL STOP AT NEXT
                                        ;BRP.  (HEAD BEGINS WRITING 1ST
```

```
                                        ;GAP BYTE.)
                                        ;
0277 76                 HLT             ;WAIT FOR INTERRUPT
                ;
                ;INTERRUPT
                ;
                ;       (EI)            ;FF GAP BYTE WRITTEN, WRITE
                                        ;CURRENT TURNS OFF.
0278 AF                 XRA     A       ;TURN OFF 372 BY RESETTING STT.
0279 D303               OUT W3
                                        ;
027B C9                 RET             ;DATA RECORD IS WRITTEN.
                        ENDIF
                ;
                ;
                ;
                ;  STEP IN
                ;  INPUT: BC=UNIT #
                ; REGISTERS: AF,D,HL
                ; STACK PAIRS: 1
                ; SUBROUTINES: DELAY,SETW0
027C 16C0       STI:    MVI D,W4STS+W4SID ;D=STROBE+"IN" DIRECTION
027E 3E01               MVI A,1         ;INCREMENT TKPTR
0280 C38702             JMP ST010
                ;
                ;  STEP OUT
                ;  INPUT: BC=UNIT #
                ; REGISTERS: AF,D,HL
                ; SUBROUTINES: DELAY,CLRW0
0283 1680       STO:    MVI     D,W4STS ;D=STROBE+"OUT" DIRECTION
0285 3EFF               MVI A,-1        ;DECREMENT TKPTR
                ;
0287 211204     ST010:  LXI     H,TKPTR
                        IF MU
028A 09                 DAD     B       ;ADD UNIT# TO HL
                        ENDIF
028B 86                 ADD M           ;INC/DEC TKPTR
028C 77                 MOV M,A
                        IF STD
                        MVI A,LHCTK     ;CHECK FOR WRITE CURRENT CHANGE
                        CMP M           ;A=LHCTK-TKPTR
                        MVI A,W0LCT
                        JM ST020        ;TRACK>LHCTK
                        CALL CLRW0      ;TRACK<OR=LHCTK,
                                        ;TURN OFF LOW CURRENT
                        XRA     A
                        JMP ST030
                ST020:  CALL SETW0      ;TRACK>LHCTK,
                                        ;TURN ON LOW CURRENT
                        MVI     A,1
                ST030:  LXI     H,LCRNT ;UPDATE LOW CURRENT
                        IF MU
                        DAD     B       ;ADD UNIT# TO HL
                        ENDIF
```

```
                        MOV     M,A     ;STORE LOW CURRENT STATUS
                        ENDIF
028D 7A                 MOV A,D
028E D304               OUT W4          ;SET DIRECTION
0290 F620               ORI W4SOS       ;TURN ON SOS
0292 D304               OUT W4          ;OUTPUT RISING EDGE OF SOS
0294 E6DF               ANI NOT W4SOS   ;TURN OFF SOS
0296 D304               OUT W4          ;OUTPUT TRAILING EDGE OF SOS
                ;  DELAY TTACS MSEC
0298 1628               MVI D,TTACS
029A CDE602             CALL DELAY
029D C9                 RET
                ;
                        IF STD
                ;
                ; HEAD LOAD SUBROUTINE
                ; INPUT: BC=UNIT#
                ; REGISTERS: AF,D,HL
                ; STACK PAIRS: 1
                ; SUBROUTINES: UASLC,DELAY,SETW0
                HDLD:   LXI H,HEAD      ;CHECK HEAD STATUS
                        IF MU
                        DAD     B       ;ADD UNIT# TO HL
                        ENDIF
                        MOV A,M         ;A=HEAD STATUS
                        ORA A           ;IS HEAD LOADED ALREADY?
                        RNZ             ;YES
                        MVI A,W0HLD     ;NO
                        CALL SETW0      ;LOAD HEAD
                        MVI D,40        ;WAIT 40 MSEC
                        CALL DELAY
                        MVI A,1         ;SET HEAD STATUS
                        JMP UL010
                ;
                ; UNLOAD HEAD SUBROUTINE
                ; INPUT: BC=UNIT#
                ; REGISTERS: AF,HL
                ; STACK PAIRS: 1
                ; SUBROUTINES: UASLC,CLRW0
                UNLD:   CALL UASLC
                        MVI A,W0HLD
                        CALL CLRW0      ;UNLOAD HEAD
                        XRA A
                UL010:  LXI H,HEAD      ;UPDATE HEAD STATUS
                        IF MU
                        DAD     B       ;ADD UNIT# TO HL
                        ENDIF
                        MOV M,A         ;STORE NEW HEAD STATUS
                        RET
                        ENDIF
                ;
                        IF MINI
                ;
                ; MOTOR ON SUBROUTINE
```

```
                ; INPUT: BC=UNIT#
                ; REGISTERS: AF,D,HL
                ; STACK PAIRS: 1
                ; SUBROUTINES: SETW0,DELAY
029E 211804     MTRON:  LXI     H,MOTOR
                        IF MU
02A1 09                 DAD     B       ;ADD UNIT# TO HL
                        ENDIF
02A2 7E                 MOV     A,M
02A3 B7                 ORA     A       ;IS MOTOR ON ALREADY?
02A4 C0                 RNZ             ;YES, RETURN
02A5 79                 MOV     A,C     ;A=UNIT #
02A6 3C                 INR     A
02A7 07                 RLC
02A8 CDD002             CALL    SETW0   ;TURN ON MOTOR
02AB 2604               MVI     H,4     ;DELAY 1 SECOND
02AD 16FA       MN010:  MVI     D,250
02AF CDE602             CALL    DELAY
02B2 25                 DCR     H       ;DONE?
02B3 C2AD02             JNZ     MN010   ;NO
02B6 3E01               MVI     A,1     ;YES
02B8 C3C202             JMP     MF010
                ;
                ; MOTOR OFF SUBROUTINE
                ; INPUT: BC=UNIT#
                ; REGISTERS: AF,HL
                ; STACK PAIRS: 1
                ; SUBROUTINES: CLRW0
02BB 79         MTOFF:  MOV     A,C     ;A=UNIT #
02BC 3C                 INR     A
02BD 07                 RLC
02BE CDC802             CALL    CLRW0   ;TURN OFF MOTOR, UNLOAD HEAD
02C1 AF                 XRA     A
02C2 211804     MF010:  LXI     H,MOTOR
                        IF MU
02C5 09                 DAD     B       ;ADD UNIT# TO HL
                        ENDIF
02C6 77                 MOV     M,A     ;UPDATE MOTOR STATUS
02C7 C9                 RET
                        ENDIF
                ;
                ;WR0 MANAGER
                ;  A=BITS TO BE CLEARED/SET
                ;  REGISTERS: AF,HL
                ; STACK PAIRS: 0
02C8 211104     CLRW0:  LXI H,WR0
02CB 2F                 CMA
02CC A6                 ANA M           ;CLEAR
02CD C3D402             JMP CR010
02D0 211104     SETW0:  LXI H,WR0
02D3 B6                 ORA M           ;SET
02D4 D300       CR010:  OUT W0
02D6 77                 MOV M,A         ;SAVE A COPY OF W0
02D7 C9                 RET
```

```
                ;
                ;
                ; UNIT A SELECT SUBROUTINE
                ;
                ; INPUT: BC=UNIT#
                ; REGISTERS: AF
                ; STACK PAIRS: Ø
Ø2D8            UASLC   EQU $
                        IF NOT MU
                        LXI     B,Ø     ;SET UNIT#=Ø
                        ENDIF
Ø2D8 79                 MOV     A,C     ;A=UNIT#
Ø2D9 3C                 INR     A
Ø2DA F6Ø4               ORI W1UAS       ;TURN ON STROBE
Ø2DC D3Ø1               OUT W1          ;SELECT UNIT
Ø2DE C9                 RET
                        IF STD
                RDYA:   IN      R1      ;CHECK FOR READY
                        ANI R1RYA       ;READY?
                        RNZ             ;YES
                        CMA             ;NO, ERROR
                        STA SLCTF       ;SET SELECT FAULT FLAG
                        JMP ERROR
                        ENDIF
                ;
                        IF MU
                ; UNIT B SELECT SUBROUTINE
                ; INPUT: C=UNIT#
                ; REGISTERS: AF
                ; STACK PAIRS: Ø
                ;
Ø2DF 79         UBSLC:  MOV     A,C
Ø2EØ 3C                 INR     A
Ø2E1 F6Ø4               ORI     W4UBS   ;TURN ON STROBE
Ø2E3 D3Ø4               OUT W4          ;SELECT UNIT
Ø2E5 C9                 RET
                        ENDIF
                ;
                ;
                ;  DELAY SUBROUTINE
                ;       D= # OF MSEC (MAX=256 WITH B=Ø)
                ;  REGISTERS: AF,D
                ;
Ø2E6 3EØ4       DELAY:  MVI A,W6TRR     ;TURN ON TRR
Ø2E8 D3Ø6               OUT W6          ;RESET TIMER REQUEST
                ;  WAIT FOR TRQ RST STATUS
Ø2EA DBØØ       DØ1Ø:   IN RØ           ;READ STATUS
Ø2EC E6Ø4               ANI RØTRQ       ;CHECK FOR TRQ
Ø2EE CAEAØ2             JZ DØ1Ø         ;WAIT FOR 1 MSEC
                        IF MINI
Ø2F1 15                 DCR     D       ;MINI CLOCK IS HALF FAST
                        ENDIF
Ø2F2 15                 DCR D           ;DONE?
Ø2F3 C2E6Ø2             JNZ DELAY       ;NO
```

```
02F6 C9                 RET             ;YES
                ;
                        IF MU
                ; DOUBLE INDEX ADRESSING SUBROUTINE
                ; INPUT: HL=BASE
                ;         BC=UNIT#
                ; OUTPUT: HL=HL+2*(UNIT#)
                ; REGISTERS: AF,HL
                ; STACK PAIRS: 0
02F7 79         DINDX:  MOV     A,C     ;A=UNIT#
02F8 07                 RLC             ;A=2*(UNIT#)
02F9 85                 ADD     L
02FA 6F                 MOV     L,A     ;L=L+2*(UNIT#)
02FB 78                 MOV     A,B
02FC 8C                 ADC     H
02FD 67                 MOV     H,A     ;H=H+B+CARRY
02FE C9                 RET             ;HL=HL+2*(UNIT#)
                        ENDIF
                ;
                        IF FMT
                ;
                ;
                ; ***** DISK FORMATTING ROUTINE *****
                ;
02FF CD2600     FRMAT:  CALL INIT       ;INITIALIZE DISK UNIT
                        IF MU
0302 CDD802             CALL    UASLC
                        ENDIF
                        IF MINI
0305 CD9E02             CALL    MTRON   ;TURN ON MOTOR, LOAD HEAD
                        ENDIF
                        IF STD
                        CALL    RDYA    ;CHECK READY STATUS
                        CALL HDLD       ;LOAD HEAD
                        ENDIF
                ; INITIALIZE ADDRESS POINTERS
0308 210100             LXI H,1         ;H=00=TRACK ADDRESS
                                        ;L=01=SECTOR ADDRESS
                ;
                ; SET UP COMMANDS
030B 3EB8       FM030:  MVI A,W1CBS+W1CBN
030D D301               OUT W1          ;SET CLOCK BITS
030F 3EFF               MVI A,0FFH
0311 D302               OUT W2          ;SET WRITE DATA=0FFH
0313 3E78               MVI A,W3WCS+W3STT+W3WES+W3IXS
0315 D303               OUT W3          ;SET 372 TO START WRITING
                                        ;AT INDEX HOLE
0317 FB                 EI              ;ENABLE INTERRUPTS AND
0318 76                 HLT             ;WAIT FOR INDEX
                ;
                ; INTERRUPT (INDEX START)
                ;
                ;       (EI)      HEAD IS WRITING FIRST GAP BYTE
```

```
0319 3E02               MVI A,W6IRR
031B D306               OUT W6          ;RESET INDEX REQUEST
                        IF STD
                ; WRITE PRE-INDEX GAP
                        MVI B,NFPRI     ; B=NUMBER OF 0FFH GAP BYTES
                FM040:  HLT             ;WAIT FOR BRP INTERRUPT
                ;
                ; INTERRUPT (DATA REQUEST)
                ;
                ;       (EI)      HEAD WRITES GAP BYTES 2-40
                        DCR B           ;DONE?
                        JNZ FM040       ; NO, REPEAT
                        XRA A           ;YES, CHANGE GAP
                        OUT W2          ;BYTE TO 00H
                        MVI B,5         ;B=BYTE COUNT
                FM050:  HLT
                ;
                ; INTERRUPT
                ;
                ;       (EI)      HEAD WRITES GAP BYTES 41-45
                        DCR B           ;DONE?
                        JNZ FM050       ; NO, REPEAT
                        HLT             ;YES
                ;
                ; INTERRUPT
                ;
                ;       (EI)      HEAD IS WRITING GAP BYTE 46
                ; WRITE INDEX ADDRESS MARK
                        MVI A,W1CBS+W1CBI
                        OUT W1          ;CHANGE CLOCK BITS
                        MVI A,0FCH
                        OUT W2          ;SET WRITE DATA=0FCH
                        HLT             ;WRITE MARK
                ;
                ; INTERRUPT
                ;
                ;       (EI)      HEAD IS WRITING INDEX ADDRESS MARK
                ; WRITE POST-INDEX GAP
                        MVI A,W1CBS+W1CBN
                        OUT W1          ;SET CLOCK BITS
                        MVI A,0FFH
                        OUT W2          ;SET WRITE DATA=0FFH
                        ENDIF
                ;
031D 060F               MVI B,NFPOI     ; B=BYTE COUNT
031F 76         FM060:  HLT             ;WRITE GAP BYTE
                ;
                ; INTERRUPT
                ;
                ;       (EI)      HEAD WRITES GAP BYTES 1 THRU NFPOI
0320 05                 DCR B           ;DONE?
0321 C21F03             JNZ FM060       ; NO, REPEAT
                ;
0324 AF         FM070:  XRA A           ;BEGINNING OF SECTOR WRITE LOOP
```

```
                                        ;-EXECUTED NSCTR TIMES
0325 D302               OUT W2          ;SET WRITE DATA=00H
0327 76                 HLT
                ;
                ; INTERRUPT
                ;
                ;       (EI)        HEAD IS WRITING 1ST 00 BYTE
0328 76                 HLT
                ;
                ; INTERRUPT
                ;
                ;       (EI)        2ND
0329 76                 HLT
                        IF STD
                ;
                ; INTERRUPT
                ;
                ;       (EI)        3RD OF 6
                        HLT
                ;
                ; INTERRUPT
                ;
                ;       (EI)        4TH OF 6
                        HLT
                        ENDIF
                ;
                ; INTERRUPT
                ;
                ;       (EI)            HEAD IS WRITING NEXT TO LAST
                                        ;FF GAP BYTE.
032A 06FE               MVI B,0FEH      ; LOAD ID MARK IN B
032C 0E22               MVI C,W3STT+W3CCG ; STORE SET CCG COMMAND IN C
                                        ; (ALSO RESETS IXS)
032E 16B8               MVI D,W1CBS+W1CBN ; STORE "FF" CLOCK PATTERN
                                        ;COMMAND IN D
0330 1E20               MVI E,W3STT     ; STORE RESET CCG COMMAND IN E
0332 3E80               MVI A,W1CBS+W1CBD ; STORE "C7" DATA MARK CLOCK
                                        ;PATTERN COMMAND IN A
0334 76                 HLT
                ;
                ; INTERRUPT
                ;
                ;       (EI)        HEAD IS WRITING LAST 00 GAP BYTE
0335 D301               OUT W1          ;SET "C7" DATA MARK CLOCK PATTERN.
0337 78                 MOV A,B         ;SET "FE" DATA BITS FOR
0338 D302               OUT W2          ;ID MARK
033A 79                 MOV A,C         ;SET CCG.  THIS CAUSES CRC
033B D303               OUT W3          ;CALCULATION TO BEGIN AT NEXT BRP.
033D 7A                 MOV A,D         ;GET "FF" DATA CLOCK BIT
                                        ;PATTERN IN A
033E 76                 HLT
                ;
                ; INTERRUPT
                ;
```

```
                ;       (EI)        HEAD IS WRITING ID ADDRESS MARK
033F D301               OUT W1              ;SET "FF" DATA CLOCK BIT PATTERN
                                            ;FOR NEXT BYTE.  HEAD NOW BEGINS
                                            ;WRITING ID MARK
0341 7B                 MOV A,E             ;RESET CCG.  (CCG MUST BE RESET
0342 D303               OUT W3              ;BEFORE NEXT BRP OR CRC CALCULATION
                                            ;WOULD BEGIN AGAIN.)
0344 7C                 MOV A,H             ;LOAD TRACK ADDRESS
0345 D302               OUT W2
0347 76                 HLT                 ;WAIT FOR INTERRUPT
                ;
                ; INTERRUPT
                ;
                ;       (EI)        HEAD IS WRITING TRACK ADDRESS
                ;
                        IF STD
                        XRA A
                        OUT W2              ;SET DATA BYTE=00H
                        HLT
                ;
                ; INTERRUPT
                ;
                ;       (EI)        HEAD IS WRITING FIRST ZERO BYTE
                        ENDIF
                ;
0348 7D                 MOV A,L
0349 D302               OUT W2              ;SET DATA BYTE=SECTOR ADDRESS
034B 76                 HLT
                ;
                ; INTERRUPT
                ;
                ;       (EI)        HEAD IS WRITING SECTOR ADDRESS
                ;
                        IF STD
                        XRA A
                        OUT W2              ;SET DATA BYTE=00H
                        HLT
                ;
                ; INTERRUPT
                ;
                ;       (EI)        HEAD IS WRITING 2ND ZERO BYTE
                        ENDIF
                ;
034C 3E21               MVI A,W3STT+W3CCW ; SET CCW. IN WRITE MODE THE 372
034E D303               OUT W3              ;WILL BEGIN WRITING BITS FROM THE
                                            ;CRC REGISTERS AT THE NEXT BRP
                                            ;FOLLOWING THE SETTING OF CCW.
0350 76                 HLT                 ;WAIT FOR INTERRUPT
                ;
                ; INTERRUPT
                ;
                ;       (EI)        HEAD IS WRITING FIRST CRC BYTE
0351 76                 HLT
                ;
```

```
                ; INTERRUPT
                ;
                ;       (EI)      HEAD IS WRITING 2ND CRC BYTE
0352 3EFF               MVI A,0FFH      ; LOAD FF GAP BYTE IN WRITE DATA
0354 D302               OUT W2          ;REGISTER
0356 3E20               MVI A,W3STT     ; RESET CCW COMMAND.  CRC BIT
0358 D303               OUT W3          ;WRITING WILL STOP AT NEXT BRP.
035A 0606               MVI B,NFGP2     ; B=BYTE COUNT
035C 76         FM080:  HLT
                ;
                ; INTERRUPT
                ;
                ;       (EI)      HEAD WRITES GAP BYTES 1 THRU NFGP2
035D 05                 DCR B           ;DONE?
035E C25C03             JNZ FM080       ; NO, REPEAT
0361 AF                 XRA A           ;YES, CHANGE GAP BYTE
0362 D302               OUT W2          ;TO 00H
0364 76                 HLT
                ;       (EI)      1ST 00
0365 76                 HLT
                ;       (EI)      2ND 00
0366 76                 HLT
                ;
                        IF STD
                ;       (EI)      NEXT 00
                        HLT
                ;       (EI)      NEXT 00
                        HLT
                        ENDIF
                ;
                ; INTERRUPT
                ;
                ;       (EI)      HEAD IS WRITING NEXT TO LAST GAP BYTE
0367 06FB               MVI B,0FBH      ; LOAD DATA MARK IN B
0369 3E80               MVI A,W1CBS+W1CBD ; STORE "C7" DATA MARK CLOCK
                                        ;PATTERN COMMAND IN A
036B 76                 HLT             ;WAIT FOR INTERRUPT
                ;
                ; INTERRUPT
                ;
                ;       (EI)      HEAD IS WRITING LAST GAP BYTE
036C D301               OUT W1          ;SET "C7" DATA MARK CLOCK PATTERN.
036E 78                 MOV A,B         ;SET "FB" DATA BITS FOR
036F D302               OUT W2          ;DATA MARK
0371 79                 MOV A,C         ;SET CCG.  THIS CAUSES CRC
0372 D303               OUT W3          ;CALCULATION TO BEGIN AT NEXT BRP.
0374 7A                 MOV A,D         ;GET "FF" DATA BIT CLOCK
                                        ;PATTERN IN A
0375 76                 HLT             ;WAIT FOR INTERRUPT.
                ;
                ; INTERRUPT
                ;
                ;       (EI)      HEAD IS WRITING DATA ADDRESS MARK
0376 D301               OUT W1          ;SET "FF" DATA BIT CLOCK PATTERN
```

```
                                        ;FOR NEXT BYTE.
0378 7B             MOV A,E             ;RESET CCG.  (CCG MUST BE RESET
0379 D303           OUT W3              ;BEFORE NEXT BRP OR CRC CALCULATION
                                        ;WOULD BEGIN AGAIN.)
037B 3EE5           MVI A,0E5H          ;LOAD DATA
037D D302           OUT W2
037F 76             HLT
                ;
                ; INTERRUPT
                ;
                ;       (EI)      DATA BYTE 1
0380 067F           MVI B,NBSCT-1
0382 76         FM100:  HLT
                ;
                ; INTERRUPT
                ;
                ;       (EI)      HEAD WRITES DATA BYTES 2-NBSCT
0383 05             DCR B
0384 C28203         JNZ FM100
                ;
0387 3E21           MVI A,W3STT+W3CCW ;SET CCW.  IN WRITE MODE THE 372
0389 D303           OUT W3              ;WILL BEGIN WRITING BITS FROM THE
                                        ;CRC REGISTERS AT THE NEXT BRP
                                        ;FOLLOWING THE NEXT SETTING OF CCW.
038B 76             HLT
                ;
                ; INTERRUPT
                ;
                ;       (EI)      HEAD IS WRITING FIRST CRC BYTE
038C 76             HLT
                ;
                ; INTERRUPT
                ;
                ;       (EI)      HEAD IS WRITING 2ND CRC BYTE
038D 3EFF           MVI A,0FFH          ;LOAD FF GAP BYTE IN WRITE DATA
038F D302           OUT W2              ;REGISTER
0391 3E20           MVI A,W3STT         ;RESET CCW COMMAND.  CRC BIT WRITING
0393 D303           OUT W3              ;ENDS
0395 0611           MVI B,NFGP3         ;B=BYTE COUNT
0397 76         FM110:  HLT
                ;
                ; INTERRUPT
                ;
                ;       (EI)      HEAD WRITES GAP BYTES 1 THRU NFGP3
0398 05             DCR B               ;DONE?
0399 C29703         JNZ FM110           ; NO, REPEAT
                ;
                ;
039C 2C             INR L               ;INCREMENT SECTOR ADDRESS
039D 3E12           MVI A,NSCTR
039F BD             CMP L               ;LAST SECTOR?
03A0 D22403         JNC     FM070       ; NO, WRITE ANOTHER SECTOR
                ;
                ; WRITE FF'S TO END OF TRACK
```

```
03A3 76        FM120:  HLT
                ;
                ; INTERRUPT
                ;
                ;       (EI)      HEAD WRITES GAP BYTES
03A4 DB00               IN RØ             ;READ STATUS
03A6 E602               ANI RØIRQ         ; INDEX REQUEST?
03A8 CAA303             JZ FM120          ; NO, CONTINUE
                ;
                ; END OF TRACK
03AB F3                 DI
03AC 3E04               MVI A,W3WER
03AE D303               OUT W3            ;WRITE ENABLE AND STT RESET.
                                          ;INDEX REQUEST IS AUTOMATICALLY RESET
                                          ;BY STT RESET.
03B0 3E22               MVI A,NTRKS-1
03B2 BC                 CMP H             ;LAST TRACK?
03B3 C8                 RZ                ; YES, FORMATTING COMPLETE
03B4 2E01               MVI L,1           ; NO, RESET SECTOR ADDRESS
03B6 24                 INR H             ;INCREMENT TRACK ADDRESS
03B7 E5                 PUSH H            ;SAVE HL
03B8 1602               MVI D,02          ;WAIT FOR TUNNEL ERASE HEAD
03BA CDE602             CALL DELAY        ;TO REACH END OF TRACK BEFORE
03BD 3A0104             LDA     UNIT
03C0 4F                 MOV     C,A
03C1 0600               MVI     B,0       ;BC=UNIT#
03C3 CD7C02             CALL STI          ;STEPPING HEAD.
                        IF MINI
03C6 160A               MVI     D,STLNG   ;HEAD SETTLING DELAY
03C8 CDE602             CALL    DELAY
                        ENDIF
03CB E1                 POP H             ;RESTORE HL
03CC C30B03             JMP FM030         ; CONTINUE
                        ENDIF
                ;
                ;
                ; ***** RAM AREA *****
                ;
03CF                    ORG 0400H
                ;COMMAND
0400            CMND:   DS 1              ;COMMAND(1-NCMDS)
                ;
                ; PARAMETERS
0401            UNIT:   DS 1              ;FDD UNIT BEING COMMANDED
0402            TRACK:  DS 1              ;TRACK DESIRED
0403            SECTR:  DS 1              ;SECTOR DESIRED
0404            SCTSZ:  DS 1              ;SECTOR SIZE
0004            NP      EQU $-UNIT        ;NO. OF PARAMETERS
                ;
                ; FLAGS
0405            MERF:   DS 1              ;MASTER ERROR
0406            CMDER:  DS 1              ;COMMAND ERROR
0407            PRMER:  DS 1              ;PARAMETER ERROR
0408            SLCTF:  DS 1              ;SELECT FAULT(NOT USED WITH MINI)
```

```
0409            NOGO:   DS 1            ;FAILED TO FIND SECTOR
040A            WTRK:   DS 1            ;WRONG TRACK
                        IF STD
                ZERO1:  DS 1            ;ZERO BYTE 1 NOT ZERO
                ZERO2:  DS 1            ;ZERO BYTE 2 NOT ZERO
                        ENDIF
040B            CRCID:  DS 1            ;CRC ERROR IN ID
040C            CRCDR:  DS 1            ;CRC ERROR IN DATA READ
040D            ILLMK:  DS 1            ;ILLEGAL DATA MARK
040E            DELMK:  DS 1            ;DELETED DATA MARK
040F            WRITF:  DS 1            ;WRITE FAULT
000B            NF      EQU $-MERF      ;NUMBER OF FLAGS
                ;
                ;COUNTERS, POINTERS, STATUSES
0410            RRTRY:  DS 1            ;READ RETRY COUNTER
0411            WR0:    DS 1            ;COPY OF LATEST W0
0412            TKPTR:  DS NU           ;TRACK POINTER FOR EACH UNIT
0414            ITIME:  DS      NU*2    ;ELAPSED IDLE TIME
                        IF STD
                HEAD:   DS NU           ;HEAD STATUS(1=LOADED, 0=UNLOADED)
                LCRNT:  DS NU           ;LOW CURRENT(1=YES,0=NO)
                        ENDIF
                        IF MINI
0418            MOTOR:  DS NU           ;MOTOR STATUS(0=OFF,1=ON)(MINI ONLY)
                        ENDIF
001A            NB      EQU $-CMND      ;NO. OF BYTES IN DATA AREA
                ;
                ;STACK
                ;
041A                    ORG 480H
                ;
                ; DATA BUFFER
0480            BUFFR:  DS NBSCT
                ;
                ;
0000                    END
```

NEC microcomputers, inc.
Five Militia Drive
Lexington, Massachusetts 02173
Telephone (617) 862-6410

PRINTED IN U.S.A.

372AD-4-77-2.5K
$5